10ª fase intermedia de ejercicios

PUNTO I
(DEFUN C:Edita (/ Mensaje)

  (SETVAR "cmdecho" 0)

  (IF ArchivoDefecto () (SETQ ArchivoDefecto ""))

  (SETQ Mensaje (STRCAT "Nombre de archivo <" ArchivoDefecto ">: "))

  (IF (= "" (SETQ Archivo (GETSTRING Mensaje)))

    (SETQ Archivo ArchivoDefecto)

  )

  (SETQ ArchivoDefecto Archivo)

  (STARTAPP "notepad" (STRCAT Archivo ".lsp"))

  (PRIN1)

)

PUNTO II
(DEFUN DatosTxArco (/ Mensaje)

  (INITGET 1)

  (SETQ Centro (GETPOINT "Centro del arco para el texto: "))(TERPRI)

  (INITGET 7)

  (SETQ Radio (GETDIST Centro "Radio del arco para el texto: "))

  (TERPRI)

  (IF AlTx0 () (SETQ AlTx0 1))

  (INITGET 6)

  (SETQ Mensaje (STRCAT "Altura del texto <" (RTOS AlTx0 2 2) ">: "))

  (IF (SETQ AlTx (GETDIST Mensaje)) () (SETQ AlTx AlTx0)) (TERPRI)

  (SETQ AlTx0 AlTx)

  (IF AngTx0 () (SETQ AngTx0 (/ PI 2)))

  (SETQ Mensaje (STRCAT "Angulo de inicio <" (ANGTOS AngTx0 0 0)

                ">: "))

  (IF (SETQ AngTx (GETANGLE Centro Mensaje)) () (SETQ AngTx AngTx0))

  (TERPRI)

  (SETQ AngTx0 AngTx)

  (SETQ CadTx (GETSTRING T "Texto: "))

)

(DEFUN Txarco (/ N LonTx AngIns Tx XIns YIns PIns AngTxN Esp)

  (SETQ LonTx (STRLEN CadTx))

  (SETQ n 1)(SETQ AngIns AngTx)

  (REPEAT Lontx

    (SETQ Tx (SUBSTR CadTx n 1))

    (SETQ XIns (+ (CAR Centro) (* Radio (COS AngIns)))) 

    (SETQ YIns (+ (CADR Centro) (* Radio (SIN AngIns))))

    (SETQ PIns (LIST XIns YIns))

    (SETQ AngTxn (/ (* (- AngIns (/ PI 2)) 180) PI))

    (COMMAND "_text" PIns AlTx AngTxN Tx)

    (SETQ Esp AlTx)

    (IF (OR (= Tx "i")(= Tx "I")(= Tx "l")(= Tx "1")(= Tx "j"))

      (SETQ Esp (/ AlTx 2))

    )

    (IF (OR (= Tx "m") (= Tx "M")) (SETQ Esp (* AlTx 1.5)))

    (SETQ AngIns (- AngIns (/ Esp Radio)))

    (SETQ N (1+ N))

  )

)

(DEFUN C:TxArco (/ Radio AlTx Centro AngTx CadTx)

  (SETVAR "cmdecho" 0)

  (SETQ Refnt0 (GETVAR "osmode"))(SETVAR "osmode" 0)

  (COMMAND "_undo" "_begin")

  (DatosTxArco)

  (TxArco)

  (COMMAND "_undo" "_end")

  (SETVAR "osmode" Refnt0)(SETVAR "cmdecho" 1)(PRIN1)

)

(PROMPT "Nuevo comando TXARCO definido.")(PRIN1)

11ª fase intermedia de ejercicios

PUNTO I
; Nuevo modo de referencia "Punto medio virtual".

; Puede ser utilizado de manera transparente entre comandos

; tecleando 'mv.

; Puede ser añadido al menu de cursor de Modos de Referencia

; de AutoCAD 14 para tenerlo a mano y no tener que teclearlo.

(DEFUN MV (/ Pto1 Pto2)

  (SETQ Pto1 (GETPOINT "medio virtual entre "))

  (SETQ Pto2 (GETPOINT Pto1 "y "))

  (POLAR Pto1 (ANGLE Pto1 Pto2) (/ (DISTANCE Pto1 Pto2) 2.0))

)

(DEFUN C:MV ()

  (mv)

)

PUNTO II
(DEFUN datos_marca (/ Mensaje)

  (IF DiaP0 () (SETQ DiaP0 2))

  (SETQ Mensaje (STRCAT "\nDiámetro del punto <" (RTOS DiaP0 2 2)

                ">: ")) 

  (INITGET 4)

  (IF (SETQ DiaP (GETDIST Mensaje)) () (SETQ DiaP DiaP0))(TERPRI)

  (SETQ DiaP0 DiaP)

  (IF DiaM0 () (SETQ DiaM0 10))

  (SETQ Mensaje (STRCAT "Diámetro del círculo de marca <"

                (RTOS DiaM0 2 2) ">: "))

  (INITGET 6)

  (IF (SETQ DiaM (GETDIST Mensaje)) () (SETQ DiaM DiaM0))(TERPRI)

  (SETQ DiaM0 DiaM)

  (INITGET 1)

  (SETQ PtI (GETPOINT "Punto inicial: "))(TERPRI)

  (INITGET 1)

  (SETQ PtF (GETPOINT PtI "Punto de situación de la marca: "))

  (TERPRI)

)  

(DEFUN marca (/ Ang Cen Mensaje Tx AlTx TxM)

  (COMMAND "_donut" "0" DiaP PtI "")

  (COMMAND "_line" PtI PtF "")

  (SETQ Ang (ANGLE PtI PtF))

  (SETQ Cen (POLAR PtF Ang (/ DiaM 2)))

  (COMMAND "_circle" Cen "_d" DiaM)

  (IF Tx0 () (SETQ Tx0 1))

  (SETQ Mensaje (STRCAT "Número de marca <" (ITOA Tx0) ">: "))

  (INITGET 6)

  (IF (SETQ Tx (GETINT Mensaje)) () (SETQ Tx Tx0))(TERPRI)

  (SETQ Tx0 (+ Tx 1))

  (SETQ AlTx (/ Diam 2))

  (SETQ TxM (ITOA Tx))

  (COMMAND "_text" "_m" Cen AlTx "0" TxM)

)

(DEFUN C:Marca (/ DiaP DiaM PtI PtF)

  (SETVAR "cmdecho" 0)

  (datos_marca)

  (SETQ refnt0 (GETVAR "osmode"))(SETVAR "osmode" 0)

  (COMMAND "_undo" "_begin")

  (marca)

  (COMMAND "_undo" "_end")

  (SETVAR "osmode" refnt0)(SETVAR "cmdecho" 1)(PRIN1)

)

(PROMPT "Nuevo comando MARCA definido.")(PRIN1)

12ª fase intermedia de ejercicios

PUNTO I
(DEFUN datos_cortetubo ()

  (SETQ pt1 (GETPOINT "\nPrimer punto del corte (modo FINal

            activado): "))

  (SETVAR "osmode" 128)

  (SETQ pt2 (GETPOINT pt1 "\nSegundo punto del corte,

            PERpendicular a: "))

  (SETVAR "osmode" 512)

  (SETQ ang2 (ANGLE pt1 (GETPOINT pt1 "\nLado de tubería,

             CERcano de: ")))

  (SETVAR "osmode" 0)

)

(DEFUN cortetubo ( / ang dis p1 p2 p3 p4 p5)

  (SETQ ang (ANGLE pt1 pt2) dis (DISTANCE pt1 pt2))

  (SETQ p1 (POLAR pt1 ang (/ dis 4)))

  (SETQ p1 (POLAR p1 (+ ang2 PI) (/ dis 6)))

  (SETQ p2 (POLAR pt1 ang (/ dis 2)))

  (SETQ p3 (POLAR p2 ang (/ dis 4)))

  (SETQ p4 (POLAR p3 (+ ang2 PI) (/ dis 6)))

  (SETQ p5 (POLAR p3 ang2 (/ dis 6)))

  (COMMAND "_arc" pt1 p1 p2)

  (COMMAND "_arc" p2 p4 pt2)

  (COMMAND "_arc" p2 p5 pt2)

  (COMMAND "_hatch" "_u" 45 2 "_n" p4 p5 "")

)

(DEFUN c:cortetubo ( / pt1 pt2 ang2)

  (SETVAR "cmdecho" 0)

  (SETQ error0 *error* *error* ControlErrores)



  (SETQ refnt0 (GETVAR "osmode"))(SETVAR "osmode" 1)

  (datos_cortetubo)

  (COMMAND "_undo" "_begin")

  (cortetubo)

  (COMMAND "_undo" "_end")

  (SETVAR "cmdecho" 1)(SETVAR "osmode" refnt0)

)

(DEFUN c:ct ()

  (c:cortetubo)

)

(DEFUN ControlErrores (CadenaError)

  (SETQ *error* error0)

  (PRINC (STRCAT "\nError: " CadenaError "."))

  (COMMAND "_undo" "_end")

  (SETVAR "osmode" refnt0)

  (SETVAR "cmdecho" 1) 

  (COMMAND)

  (TERPRI)

  (PRIN1)

)

(PROMPT "\nNuevo comando CORTETUBO (abreviatura CT) creado.")(PRIN1)

PUNTOII
(DEFUN datos_puerta ( / ptang dist mens)

  (SETVAR "osmode" 1)

  (INITGET 1)

  (SETQ ptb (GETPOINT "\nSeñalar esquina (FINal de): "))

  (SETVAR "osmode" 512)

  (INITGET 1)

  (SETQ ptang (GETPOINT ptb "\nSeñalar pared donde abrir el hueco

              (CERca de): "))

  (SETQ ang (ANGLE ptb ptang))

  (SETVAR "osmode" 0)

  (INITGET 1)

  (SETQ dist (GETDIST ptb "\nDistancia desde la esquina: "))

  (SETQ ptins (POLAR ptb ang dist))

  (IF marco0 ( ) (SETQ marco0 5))

  (SETQ mens (STRCAT "\nEspesor del marco <" (RTOS marco0 2 2)

             ">: "))

  (INITGET 6)

  (IF (SETQ marco (GETDIST ptins mens)) () (SETQ marco marco0))

  (SETQ marco0 marco)

  (IF hoja0 ( ) (SETQ hoja0 5))

  (SETQ mens (STRCAT "\nEspesor de la hoja <" (RTOS hoja0 2 2)

             ">: "))

  (INITGET 6)

  (IF (SETQ hoja (GETDIST mens)) () (SETQ hoja hoja0))

  (SETQ hoja0 hoja)

  (IF ancho0 () (SETQ ancho0 90))

  (SETQ mens (STRCAT "\nAnchura del hueco de la puerta <"

             (RTOS ancho0 2 2) ">: "))

  (INITGET 6)

  (IF (SETQ ancho (GETDIST ptins mens)) () (SETQ ancho ancho0))

  (SETQ ancho0 ancho)

  (SETVAR "osmode" 128)

  (INITGET 1)

  (SETQ ptlado (GETPOINT ptins "\nSeñalar la pared opuesta del muro

               o tabique (PER a): "))

  (SETVAR "osmode" 0)

  (INITGET "Actual Opuesta")

  (IF (SETQ op (GETKWORD "\nDirección de apertura (Actual/Opuesta)

               <A>: ")) () (SETQ op "Actual"))

)

(DEFUN hueco ()

  (SETQ ptop (POLAR ptins ang ancho))

  (SETQ ptlop (POLAR ptlado ang ancho))

  (SETVAR "pickbox" 1)

  (COMMAND "_break" ptins ptop)

  (COMMAND "_break" ptlado ptlop)

  (COMMAND "_line" ptins ptlado "")

  (COMMAND "_matchprop" ptb "_l" "")

  (COMMAND "_line" ptop ptlop "")

  (COMMAND "_matchprop" ptb "_l" "")

)

(DEFUN puerta ( / pt pt1 pt2 pt3 pt4 ptarc lhoja)

  (IF (= op "Actual")

    (PROGN 

      (SETQ pt ptins ptins ptop ptop pt)

      (SETQ pt ptlado ptlado ptlop ptlop pt)

      (SETQ ang (- ang PI))

     )

  )

  (SETQ pt1 (POLAR ptins (ANGLE ptins ptlado) marco))

  (SETQ pt2 (POLAR pt1 ang marco))

  (SETQ pt3 (POLAR pt2 (ANGLE ptlado ptins) marco))

  (SETQ ptarc pt3)

  (COMMAND "_line" pt1 pt2 pt3 ptins "")

  (SETQ pt1 (POLAR ptop (ANGLE ptins ptlado) marco))

  (SETQ pt2 (POLAR pt1 (- ang PI) marco))

  (SETQ pt3 (POLAR pt2 (ANGLE ptlado ptins) marco))

  (COMMAND "_line" pt1 pt2 pt3 ptop "")

  (SETQ lhoja (- ancho marco marco))

  (SETQ pt2 (POLAR pt3 (- ang PI) hoja))

  (SETQ pt1 (POLAR pt2 (ANGLE ptlado ptins) lhoja))

  (SETQ pt4 (POLAR pt1 ang hoja))

  (COMMAND "_line" pt3 pt2 pt1 pt4 pt3 "")

  (SETQ angarc (- (ANGLE pt2 ptarc) (ANGLE pt2 pt1) ))

  (SETQ angarc (/ (* angarc 180) PI))

  (IF (>= angarc 180) (SETQ angarc (- angarc 360)))

  (IF (<= angarc -180) (SETQ angarc (+ angarc 360)))

  (COMMAND "_arc" "_c" pt3 pt4 "_a" angarc)

)

(DEFUN c:puerta ( / ptb ang ptins marco hoja ancho ptlado op ptop

                ptlop)

  (SETVAR "cmdecho" 0)

  (SETQ error0 *error* *error* ControlErrores)

  (SETQ refnt0 (GETVAR "osmode"))(SETQ pick0 (GETVAR "pickbox"))

  (datos_puerta)

  (COMMAND "_undo" "_begin")

  (hueco)

  (puerta)

  (COMMAND "_undo" "_end")

  (SETVAR "cmdecho" 1)(SETVAR "osmode" refnt0)(SETVAR "pickbox"

                      pick0)(PRIN1)

)

(DEFUN ControlErrores (CadenaError)

  (SETQ *error* error0)

  (PRINC (STRCAT "\nError: " CadenaError "."))

  (COMMAND "_undo" "_end")

  (SETVAR "pickbox" pick0) (SETVAR "osmode" refnt0)

  (SETVAR "cmdecho" 1) 

  (COMMAND)

  (TERPRI)

  (PRIN1)

)

(PROMPT "\nNuevo comando PUERTA definido.")(PRIN1)

13ª fase intermedia de ejercicios

PUNTO I
;;; Esto es un archivo ACAD.LSP personalizado.

;; Las siguientes funciones serán cargadas en memoria al comenzar

(DEFUN C:...

...

)

;; Las siguientes rutinas AutoLISP, ADS y ARX serán evaluadas en el inicio

(LOAD "c:\\...")

(LOAD "c:\\...")

(LOAD "d:\\...")

(XLOAD "c:\\...")

(XLOAD "d:\\...")

(ARXLOAD "c:\\...")

(ARXLOAD "c:\\...")

;; Las siguientes rutinas AutoLISP, ADS y ARX serán cargadas y

;; evaluadas al ser llamadas

(AUTOLOAD "c:/..." '("..."))

(AUTOLOAD "c:/..." '("..."))

(AUTOLOAD "d:/..." '("..."))

(AUTOLOAD "a:/..." '("..."))

(AUTOXLOAD "c:/..." '("..."))

(AUTOXLOAD "c:/..." '("..."))

(AUTOARXLOAD "c:/..." '("..."))

(AUTOARXLOAD "c:/..." '("..."))

;; Lo que sigue será evaluado directamente al comenzar

(DEFUN S::STARTUP ()

...

)

;; Lo que sigue también será evaluado directamente al comenzar

(...)

;;; Fin del archivo ACAD.LSP personalizado.

14ª fase intermedia de ejercicios

PUNTO I
Letrero a)

(DEFUN defecto_agujeros ()

  (START_IMAGE "imagen")

  (SETQ animg (DIMX_TILE "imagen") altimg (DIMY_TILE "imagen"))

  (FILL_IMAGE 0 0 animg altimg -2)

  (SLIDE_IMAGE 0 0 animg altimg "sin.sld")

  (END_IMAGE)

  (SET_TILE "sin" "1")

  (SETQ op "sin")

  (IF diamagu () (SETQ diamagu 1))

  (SET_TILE "diamagu" (RTOS diamagu 2 2))

  (IF profagu () (SETQ profagu 1))

  (SET_TILE "profagu" (RTOS profagu 2 2))

  (IF diamcaj () (SETQ diamcaj 1))

  (SET_TILE "diamcaj" (RTOS diamcaj 2 2))

  (IF profcaj () (SETQ profcaj 1))

  (SET_TILE "profcaj" (RTOS profcaj 2 2))

  (MODE_TILE "cajera" 1)

)

(DEFUN acciones_cuadro ()

  (ACTION_TILE "tipo" "(tipo)")

  (ACTION_TILE "accept" "(aceptar_agujeros)") 

)

(DEFUN tipo  ()

  (SETQ op $value)

  (IF (= op "sin") (MODE_TILE "cajera" 1) (MODE_TILE "cajera" 0))

  (COND ((= op "sin") (START_IMAGE "imagen")(FILL_IMAGE 0 0 animg

                           altimg -2)

    

 (SLIDE_IMAGE 0 0 animg altimg "sin.sld")

             (END_IMAGE))

        ((= op "recta") (START_IMAGE "imagen")(FILL_IMAGE 0 0 animg

                        


     altimg -2)

             (SLIDE_IMAGE 0 0 animg altimg "recta.sld")

             (END_IMAGE))

        ((= op "avella") (START_IMAGE "imagen")(FILL_IMAGE 0 0 animg

      altimg -2)

             (SLIDE_IMAGE 0 0 animg altimg "avella.sld")

             (END_IMAGE))

   )

)

(DEFUN aceptar_agujeros ( )

  (SETQ diamagu (ATOF (GET_TILE "diamagu")))

  (SETQ profagu (ATOF (GET_TILE "profagu")))

  (SETQ diamcaj (ATOF (GET_TILE "diamcaj")))

  (SETQ profcaj (ATOF (GET_TILE "profcaj")))

  (COND ((<= diamagu 0)(MODE_TILE "diamagu" 2)

     (SET_TILE "error" "Diámetro de agujero debe ser mayor que 0"))

     ((<= profagu 0)(MODE_TILE "profagu" 2)

     (SET_TILE "error" "Profundidad de agujero debe ser mayor que 0"))

     ((AND (<= diamcaj 0) (/= op "sin"))(MODE_TILE "diamcaj" 2)

     (SET_TILE "error" "Diámetro de cajera debe ser mayor que 0"))

     ((AND (<= diamcaj diamagu) (/= op "sin"))(MODE_TILE "diamcaj" 2)

     (SET_TILE "error" "Diámetro de cajera debe ser mayor que agujero

     "))

     ((AND (<= profcaj 0) (/= op "sin"))(MODE_TILE "profcaj" 2)

     (SET_TILE "error" "Profundidad de cajera debe ser mayor que 0"))

     (T (DONE_DIALOG 1))

   )

)

(DEFUN insertar_agujeros ( / mens pticj ptigj ang)

  (INITGET 1)

  (SETQ pticj (GETPOINT "Punto de inserción: ")) (TERPRI)

  (IF ang0 () (SETQ ang0 0))

  (SETQ mens (STRCAT "Angulo para la inserción <" (ANGTOS ang0 0 2)

 ">: ")) 

  (IF (SETQ ang (GETANGLE pticj mens)) () (SETQ ang ang0)) (TERPRI)

  (SETQ ang0 ang)

  (IF (/= op "sin")(cajera)(SETQ ptigj pticj))

  (agujero)

  (eje)

)

(DEFUN cajera ( / pt1 pt2 pt3 pt4)

  (SETQ pt1 (POLAR pticj (- ang (/ PI 2)) (/ diamcaj 2)))

  (SETQ pt2 (POLAR pt1 ang profcaj))

  (SETQ pt3 (POLAR pt2 (+ ang (/ PI 2)) diamcaj))

  (SETQ pt4 (POLAR pt3 (+ ang PI) profcaj))

  (IF (= op "avella") (PROGN (SETQ pt1 (POLAR pt1 (- ang (/ PI 2))

profcaj))

                             (SETQ pt4 (POLAR pt4 (+ ang (/ PI 2))

profcaj)))

  )

  (COMMAND "_line" pt1 pt2 pt3 pt4 "")

  (SETQ ptigj (POLAR pticj ang profcaj))

)

(DEFUN agujero ( / pt1 pt2 pt3 pt4 pt5)

  (SETQ pt1 (POLAR ptigj (- ang (/ PI 2)) (/ diamagu 2)))

  (SETQ pt2 (POLAR pt1 ang profagu))

  (SETQ pt3 (POLAR pt2 (+ ang (/ PI 2)) (/ diamagu 2)))

  (SETQ pt3 (POLAR pt3 ang (/ diamagu 2)))

  (SETQ pt4 (POLAR pt2 (+ ang (/ PI 2)) diamagu))

  (SETQ pt5 (POLAR pt4 (+ ang PI) profagu))

  (COMMAND "_line" pt1 pt2 pt4 pt5 "")

  (COMMAND "_line" pt2 pt3 pt4 "")

)

(DEFUN eje (/ lon lonj ptij capj ptfj)

  (SETQ lon (+ profcaj profagu (/ diamagu 2)))

  (SETQ lonj (* lon 1.25))

  (SETQ ptij (POLAR pticj (+ ang PI) (* lon 0.125)))

  (IF (= "" (SETQ capj (GETSTRING "Nombre de capa para el eje <EJES>:

  ")))

            (SETQ capj "EJES")

  )(TERPRI)

  (SETQ ptfj (POLAR ptij ang lonj))

  (COMMAND "_layer" "_make" capj "")

  (COMMAND "_line" ptij ptfj "")

  (COMMAND "_layer" "_set" cap0 "")

)

(DEFUN c:ddagujeros ( / animg altimg op)

  (SETVAR "cmdecho" 0)

  (SETQ error0 *error* *error* errores)

  (SETQ ind (LOAD_DIALOG "agujeros.dcl"))

  (IF (NOT (NEW_DIALOG "agujeros" ind))

    (PROGN (PROMPT "\nNo se encuentra el archivo AGUJEROS.DCL")(QUIT))

  )

  (defecto_agujeros)

  (acciones_cuadro)

  (IF (= 1 (START_DIALOG))(insertar_agujeros))

  (UNLOAD_DIALOG ind)

  (SETVAR "cmdecho" 1)(PRIN1)

)

(DEFUN errores (mens)

  (SETQ *error* error0)

  (IF (= mens "quitar / salir abandonar")

    (PRIN1)

    (PRINC (STRCAT "\nError: " mens " "))

  )

  (SETVAR "cmdecho" 1)(PRIN1)

)

(PROMPT "Nuevo comando DDAGUJEROS definido")(PRIN1)


NOTA: Apréciese la utilización de diversos códigos de estado con DONE_DIALOG para ser luego controlados en el START_DIALOG.

Letreros b-1 y b-2)

(DEFUN listabl_inserdir (/ arch línea)

  (IF ruta0 () (SETQ ruta0 "c:\\cad13"))

  (IF (= "" (SETQ ruta (GETSTRING (STRCAT "\nDirectorio de bloques <" 

 ruta0 ">: "))))

            (SETQ ruta ruta0))

  (WHILE (NOT (FINDFILE (STRCAT ruta "\\bloques.lst")))

    (PROMPT "\nNo se ha encontrado archivo BLOQUES.LST con listado de 

    archivos de bloque"))

    (IF (SETQ ruta (GETSTRING (STRCAT "\nDirectorio de bloques <" 

    ruta0 ">: ")))

      () (SETQ ruta ruta0))

  )

  (SETQ ruta0 ruta)

  (SETQ arch (OPEN (STRCAT ruta "\\bloques.lst") "r"))

  (SETQ listabl '())

  (WHILE (SETQ línea (READ-LINE arch))

    (SETQ nombl (SUBSTR línea 1 (- (STRLEN línea) 4)))

    (IF (/= nombl "")(SETQ listabl (CONS nombl listabl)))

  )

  (CLOSE arch)

  (SETQ listabl (REVERSE listabl))

  (SETQ totbl (LENGTH listabl))

)   

(DEFUN cuadro (/ nimg totimg nbl nombl nomtile)

  (IF (<= restbl 16)(SETQ totimg restbl)(SETQ totimg 16))

  (SETQ nimg 0)

  (REPEAT totimg

    (SETQ nbl (+ indbl nimg))

    (SETQ nombl (STRCAT ruta "\\" (NTH nbl listabl)))

    (SETQ nomtile (STRCAT "img" (ITOA nimg)))

    (START_IMAGE nomtile)

    (FILL_IMAGE 0 0 (DIMX_TILE nomtile) (DIMY_TILE nomtile) 0)

    (SLIDE_IMAGE 0 0 (DIMX_TILE nomtile) (DIMY_TILE nomtile) nombl)

    (END_IMAGE)

    (SETQ nimg (1+ nimg))

  )

  (REPEAT (- 16 totimg)

    (SETQ nbl (+ indbl nimg))

    (SETQ nomtile (STRCAT "img" (ITOA nimg)))

    (START_IMAGE nomtile)

    (FILL_IMAGE 0 0 (DIMX_TILE nomtile) (DIMY_TILE nomtile) 0)

    (END_IMAGE)

    (SETQ nimg (1+ nimg))

  )

  (IF (= indbl 0) (MODE_TILE "pre" 1) (MODE_TILE "pre" 0))

  (IF (<= restbl 16) (MODE_TILE "sig" 1) (MODE_TILE "sig" 0))

)

(DEFUN lista (/ indblst)

  (SETQ nombl (NTH nblist listabl))

  (IF (= ins 4) (DONE_DIALOG 1))

  (SETQ indblst (* 16 (FIX (/ nblist 16))))

  (SETQ restbl (- totbl indblst))

  (IF (/= indblst indbl) (PROGN (SETQ indbl indblst)(cuadro)) )

  (SETQ nimg (- nblist indbl))

  (MODE_TILE (STRCAT "img" (ITOA nimg)) 2)

)

(DEFUN mostrar ( )

  (NEW_DIALOG "muestra" ind)

  (START_IMAGE "muestra")

  (FILL_IMAGE 0 0 (DIMX_TILE "muestra") (DIMY_TILE "muestra") 0)

  (SLIDE_IMAGE 0 0 (DIMX_TILE "muestra") (DIMY_TILE "muestra")(STRCAT 

ruta "\\" nombl))

  (END_IMAGE)

  (START_DIALOG)

  (MODE_TILE (STRCAT "img" (ITOA nimg)) 2)

)

(DEFUN funciones_img (/ nfun nomfun)

  (SETQ nfun 0)

  (REPEAT 16

    (SETQ nomfun (STRCAT "img" (ITOA nfun)))

    (EVAL (READ (STRCAT "(defun " nomfun " () "

                          "(setq nimg " (ITOA nfun) ")"

                          "(setq nbl (+ " (ITOA nfun) " indbl))"

                          "(setq nombl  (nth  nbl listabl))"

                          "(if (= ins 4) (done_dialog 1))"

                          "(set_tile \"lista\" (itoa nbl)))"  )))

    (SETQ nfun (1+ nfun))

  )

)

(DEFUN sig ()

  (SETQ indbl (+ indbl 16) restbl (- restbl 16))

  (cuadro)

  (SET_TILE "lista" (ITOA indbl))

  (MODE_TILE "img0" 2)

  (SETQ nombl (NTH indbl listabl))

  (SETQ nimg 0)

)

(DEFUN pre ()

  (SETQ indbl (- indbl 16) restbl (+ restbl 16))

  (cuadro)

  (SET_TILE "lista" (ITOA indbl))

  (MODE_TILE "img0" 2)

  (SETQ nombl (NTH indbl listabl))

  (SETQ nimg 0)

)

(DEFUN c:ddinserdir (/ ruta listabl totbl indbl restbl nblist ins ind 

  nimg nombl nomtile)

  (SETVAR "cmdecho" 0)(SETQ error0 *error* *error* err_ddinserdir)

  (listabl_inserdir)

  (SETQ indbl 0 restbl totbl)

  (SETQ ind (LOAD_DIALOG "inserdir"))

  (NEW_DIALOG "inserdir" ind)

  (START_LIST "lista")

  (MAPCAR 'add_list listabl)

  (END_LIST)

  (cuadro)

  (SET_TILE "lista" (ITOA indbl))

  (MODE_TILE "img0" 2)

  (SETQ nombl (NTH 0 listabl))

  (ACTION_TILE "lista" "(setq nblist (ATOI $value))(setq ins  

  $reason)(lista)"))

  (ACTION_TILE "mostrar" "(if nombl (mostrar))")

  (SETQ nimg 0)

  (REPEAT 16

    (SETQ nomtile (STRCAT "img" (ITOA nimg)))

    (ACTION_TILE nomtile (STRCAT "(setq ins $reason)(" nomtile ")"))

    (SETQ nimg (1+ nimg))

  )

  (ACTION_TILE "sig" "(sig)")

  (ACTION_TILE "pre" "(pre)")

  (funciones_img)

  (IF (= 1 (START_DIALOG)) 

    (PROGN (PROMPT "Punto de inserción: ")

    (COMMAND "insert" (STRCAT ruta "\\" nombl) pause "" ""))

   )

   (SETVAR "cmdecho" 1)(SETQ *error* error0)(PRIN1)

)

(DEFUN err_ddinserdir (mens)

  (SETQ *error* error0)

  (IF (= mens "quitar / salir abandonar")

    (PRIN1)

    (PRINC (STRCAT "\nError: " mens " "))

  )

  (SETVAR "cmdecho" 1)(SETQ *error* error0)(PRIN1)

)

(PROMPT "Nuevo comando DDINSERDIR definido.")(PRIN1)

Letrero c)

(DEFUN c:ddsombra (/error0 ind edge muestra)

  (SETQ error0 *error* *error* err-ddmodelar)

  (IF esfr0 () (SETQ esfr0 "1"))

  (SETQ ind (LOAD_DIALOG "modelar"))

  (IF (NOT (NEW_DIALOG "modelar" ind))

    (PROGN (PROMPT "No se encuentra el archivo MODELAR.DCL")(QUIT))

  )

  (COND

    ((= 0 (GETVAR "shadedge"))

      (SET_TILE "edge0" "1")

      (SETQ edge "edge0"))

    ((= 1 (GETVAR "shadedge"))

      (SET_TILE "edge1" "1")

      (SETQ edge "edge1"))

    ((= 2 (GETVAR "shadedge"))

      (SET_TILE "edge2" "1")

      (SETQ edge "edge2"))

    ((= 3 (GETVAR "shadedge"))

      (SET_TILE "edge3" "1")

      (SETQ edge "edge3"))

  )

  (COND

    ((= (GETVAR "shadedif") 30)(SET_TILE "dif30" "1"))

    ((= (GETVAR "shadedif") 50)(SET_TILE "dif50" "1"))

    ((= (GETVAR "shadedif") 70)(SET_TILE "dif70" "1"))

    ((= (GETVAR "shadedif") 90)(SET_TILE "dif90" "1"))

  )

  (SET_TILE "desliza" (ITOA (GETVAR "shadedif")))

  (SET_TILE "valor" (ITOA (GETVAR "shadedif")))

  (SET_TILE "esfera" esfr0)

  (SET_TILE "cubo" (ITOA (1- (ATOI esfr0))))

  (IF (= "1" esfr0) (SETQ muestra "esfera") (SETQ muestra "cubo))

  (ACTION_TILE "radio_edge" "(SETQ edge $value)")

  (ACTION_TILE "defecto" "(defecto)")

  (ACTION_TILE "desliza" "(desliza)")

  (ACTION_TILE "valor" "(valor)")

  (ACTION_TILE "dif30" "(valorp \"30\")")

  (ACTION_TILE "dif50" "(valorp \"50\")")

  (ACTION_TILE "dif90" "(valorp \"70\")")

  (ACTION_TILE "dif90" "(valorp \"90\")")

  (ACTION_TILE "muestra" "(SETQ muestra $value)")

  (ACTION_TILE "muestra" "(mostrar)")

  (ACTION_TILE "accept" "(aceptar)")

  (IF (= 1 (START_DIALOG)) (COMMAND "sombra"))

  (SETQ *error* error0) (PRIN1)

)

(DEFUN aceptar (/val)

  (SETQ val (GET_TILE "valor"))

  (IF (AND (<= (ATOI val) 100) (>= (ATOI val) 0))

    (PROGN

      (SETVAR "shadedge" (ATOI (SUBSTR edge 5)))

      (SETVAR "shadeif" (ATOI val))

      (SETQ esfr0 (GET_TILE "esfera"))

      (DONE_DIALOG 1)

    )

    (MODE_TILE "valor" 2)

  )

)

(DEFUN mostrar (/ foto ancho alto)

  (SETQ foto (STRCAT (SUBSTR muestra 1 3)(SUBSTR edge 5)))

  (SETQ ancho (DIMX_TILE "imagen") alto DIMY_TILE "imagen"))

  (START_IMAGE "imagen")

  (FILL_IMAGE 0 0 ancho alto -2)

  (SLIDE_IMAGE 0 0 ancho alto foto)

  (END_IMAGE)

)

(DEFUN defecto ()

  (SET_TILE "edge" "1")

  (SET_TILE "dif70" "1")

  (SET_TILE "desliza" "70")

  (SET_TILE "valor" "70")(SET_TILE "error" "")

  (SET_TILE "esfera" "1")

  (SETQ edge "edge1" muestra "esfera")

)

(DEFUN desliza ()

  (SET_TILE "valor" (GET_TILE "desliza"))

  (FOREACH p '("dif30" "dif50" "dif70" "dif90") (SET_TILE p "0"))

  (SET_TILE "error" "")

)

(DEFUN valor (/ val)

  (SETQ val (GET_TILE "valor"))

  IF (AND (<= (ATOI val) 100)(>= (ATOI val 0))

    (PROGN

      (SET_TILE "error" "")

      (SET_TILE "desliza" val)

      (FOREACH p '("dif30" "dif50" "dif70" "dif90") (SET_TILE p "0"))

     ) 

     (SET_TILE "error" "El valor debe estar entre 0 y 100")

  )

)

(DEFUN valorp (p)

  (SET_TILE "valor" p) (SET_TILE "desliza" p) (SET_TILE "error" "")

)

(DEFUN err-ddmodelar (mens)

  (SETQ *error* error0)

  (IF (= mens "quitar / salir abandonar")

    (PRIN1)

    (PRINC (STRCAT "\nError: " mens " "))

  )

  (PRIN1)

)

(PROMPT "Nueva orden DDMODELAR definida.")(PRIN1)


NOTA: Apréciese la manera de incluir caracteres de comillas dobles con el código de control correspondiente \"; como por ejemplo en "(valorp \"30\")". Esto se hace así por estar incluidas dentro de una cadena —que lleva ya comillas de por sí—.


NOTA: En estos ejercicios que hemos visto los archivos de fotos (.SLD) que se representen en casillas de imagen habrán de estar en un directorio de soporte. También pueden situarse en cualquier otro directorio, pero habremos de tomar la precaución de incluir la ruta en la llamada al fichero en cuestión. Lo mismo para los archivos de definición del diseño gráfico del letrero de diálogo (.DCL).
